Regression Models for Count Outcomes Using SAS
(commands= finan_count.sas)

The examples in this handout consider the TECUMSEH data set in the SASDATA2 archive (see the information provided on this data set for more details). To demonstrate how to fit generalized linear models to non-normal outcomes that represent counts, we will consider several variables measured in Round 1 of the Tecumseh study. Specifically, we demonstrate how to fit a regression model to the count variable representing the number of glasses of beer per day when beer was consumed (BEER1).
First, we consider descriptive statistics for the BEER1 variable in SAS:

libname sasdata2 "C:\temp\sasdata2";
/* descriptive statistics for BEER1 */

proc freq data = sasdata2.tecumseh;

   tables beer1;

run;

proc means data = sasdata2.tecumseh n nmiss mean var;

   var beer1;

run;
                                     The FREQ Procedure

                                  BEER, NUMBER OF GLASSES CVI

                                                    Cumulative    Cumulative

                  BEER1    Frequency     Percent     Frequency      Percent

                  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

                      0        2478       72.14          2478        72.14

                      1         408       11.88          2886        84.02

                      2         249        7.25          3135        91.27

                      3         111        3.23          3246        94.50

                      4          79        2.30          3325        96.80

                      5          36        1.05          3361        97.85

                      6          33        0.96          3394        98.81

                      7          12        0.35          3406        99.16

                      8           7        0.20          3413        99.36

                      9           9        0.26          3422        99.62

                     10           1        0.03          3423        99.65

                     12           4        0.12          3427        99.77

                     13           2        0.06          3429        99.83

                     14           3        0.09          3432        99.91

                     17           1        0.03          3433        99.94

                     18           1        0.03          3434        99.97

                     24           1        0.03          3435       100.00

                                   Frequency Missing = 1250

                                    The MEANS Procedure

                     Analysis Variable : BEER1 BEER, NUMBER OF GLASSES CVI

                                    N

                            N    Miss            Mean        Variance

                         --------------------------------------------

                         3435    1250       0.6809316       2.5388180

                         --------------------------------------------
Note the large number of missing cases for this particular variable (1,250). Who might these cases be? In addition, note that the mean of the BEER1 variable is smaller than the variance of the variable. This has important implications for the type of model that we will fit to the data.
We also investigate a histogram and a Q-Q plot for the BEER1 variable using PROC UNIVARIATE:

proc univariate data = sasdata2.tecumseh;

   histogram;

   qqplot / normal(mu = est sigma = est);

   var beer1;

run;

The BEER1 variable clearly does not appear to follow a normal distribution, and we would not expect normality, because the outcome variable represents a count (with several zeroes). We therefore consider a generalized linear model for the BEER1 outcome, with a distribution for the outcome variable that is appropriate for count data. Two commonly used distributions for count outcomes are the Poisson distribution and the Negative Binomial distribution.

We first fit a Poisson regression model to the BEER1 outcome using Proc Genmod in SAS. As predictors of the number of beers consumed, we consider SEX, AGE1, MARITAL1, ED1, CIG1, and WTKG1. An important characteristic of a Poisson random variable is that the mean is equal to the variance. Based on the initial descriptive analysis of the BEER1 variable, we have some reason to believe that the Poisson distribution may not be appropriate. Here is the SAS code to fit the model:
/* Fit Poisson Regression model, predicting BEER1 with SEX, AGE1, 

MARITAL1, ED1, CIG1, and WTKG1 */

proc genmod data = sasdata2.tecumseh;

   class sex marital1;

   model beer1 = sex age1 marital1 ed1 cig1 wtkg1 / 


dist = poisson link = log type3;

run;
Note the additional options in the model statement in Proc Genmod. The distribution of the outcome variable is specified as poisson, and the “link” function is specified as log (this is the default “canonical” link), which is how the Poisson regression model is commonly specified. The natural log of the Poisson-distributed dependent variable (BEER1) is defined by a linear combination of predictor variables. The TYPE3 statement requests overall Type III likelihood ratio tests for the predictors in the model (similar to F-tests in an analysis of variance setting), which allow analysts to get an idea of whether the predictors (categorical or continuous) are significant or not.
Here is selected output from the model fit:

                                     Model Information

              Data Set            SASDATA2.TECUMSEH

              Distribution                  Poisson

              Link Function                     Log

              Dependent Variable              BEER1    BEER, NUMBER OF

                                                       GLASSES CVI

                            Number of Observations Read        4685

                            Number of Observations Used        3275

                            Missing Values                     1410

                           Criteria For Assessing Goodness Of Fit

                 Criterion                 DF           Value        Value/DF

                 Deviance                3265       6479.9078          1.9847

                 Scaled Deviance         3265       6479.9078          1.9847

                 Pearson Chi-Square      3265       9930.5282          3.0415

                 Scaled Pearson X2       3265       9930.5282          3.0415

                 Log Likelihood                    -2897.7313

          Algorithm converged.

                                Analysis Of Parameter Estimates

                                     Standard     Wald 95% Confidence       Chi-

 Parameter         DF    Estimate       Error           Limits            Square    Pr > ChiSq

 Intercept          1     -1.9657      0.7203     -3.3775     -0.5539       7.45        0.0064

 SEX          1     1      0.5558      0.0539      0.4501      0.6616     106.18        <.0001

 SEX          2     0      0.0000      0.0000      0.0000      0.0000        .           .

 AGE1               1     -0.0051      0.0018     -0.0086     -0.0016       8.08        0.0045

 MARITAL1     1     1      1.6486      0.7077      0.2615      3.0357       5.43        0.0198

 MARITAL1     2     1      1.8850      0.7129      0.4879      3.2822       6.99        0.0082

 MARITAL1     3     1      1.6329      0.7210      0.2197      3.0461       5.13        0.0235

 MARITAL1     4     1      1.7297      0.7204      0.3177      3.1416       5.76        0.0163

 MARITAL1     5     0      0.0000      0.0000      0.0000      0.0000        .           .

 ED1                1     -0.2734      0.0296     -0.3314     -0.2154      85.31        <.0001

 CIG1               1      0.2933      0.0474      0.2004      0.3863      38.23        <.0001

 WTKG1              1      0.0007      0.0017     -0.0025      0.0040       0.19        0.6592

 Scale              0      1.0000      0.0000      1.0000      1.0000

                             LR Statistics For Type 3 Analysis

                                                  Chi-

                         Source           DF     Square    Pr > ChiSq

                         SEX               1     111.20        <.0001

                         AGE1              1       8.18        0.0042

                         MARITAL1          4      17.00        0.0019

                         ED1               1      88.98        <.0001

                         CIG1              1      39.41        <.0001

                         WTKG1             1       0.19        0.6598
The likelihood ratio test statistics allow one to test the null hypothesis that including a given predictor in a model is not improving the fit of the model. We see evidence of WTKG1 (weight) not being a significant predictor of the number of beers consumed. 

A good diagnostic check of whether or not the Poisson distribution is a good fit for this count outcome is to investigate the Value/DF value for the model Deviance (something like the sum of squares for non-normal outcomes). Basically, a model with a good fit to the data will have a Value/DF value close to or below 1, and we see that this value in this case is nearly 2, suggesting a somewhat poor fit. We noticed in the initial data summary that the variance of the count outcome was larger than the mean, and these types of count variables are often referred to as “over-dispersed” count variables as a result. A distribution often used for count outcomes with this characteristic is the negative binomial distribution, which can be specified by changing the dist option in Proc Genmod to be negbin. We again use the log link.
/* Fit Negative Binomial Regression model, predicting BEER1 with SEX, AGE1, MARITAL1, ED1, CIG1, and WTKG1 */

proc genmod data = sasdata2.tecumseh;

   class sex marital1;

   model beer1 = sex age1 marital1 ed1 cig1 wtkg1 / 


dist = negbin link = log type3;

run;
We assess the goodness of fit criteria again:
                           Criteria For Assessing Goodness Of Fit

                 Criterion                 DF           Value        Value/DF

                 Deviance                3265       2169.9814          0.6646

                 Scaled Deviance         3265       2169.9814          0.6646

                 Pearson Chi-Square      3265       2926.0569          0.8962

                 Scaled Pearson X2       3265       2926.0569          0.8962

                 Log Likelihood                    -1864.6996
Note the marked improvement in the Value/DF criterion for the model Deviance. We now have reason to believe that this model fits the data well (or at least better than the Poisson distribution did).

We consider interpretation of the estimated coefficients in the model at this point, but we have to keep in mind that we are using a natural log link function to define the generalized linear model. When fitting regression models to count data with log links, it is helpful to exponentiate the estimated coefficients. The resulting values represent ratios of expected counts on the dependent variable associated with a one-unit increase in a given predictor. In other words, what is the expected multiplicative increase in the mean of the count response that is associated with a one-unit increase in a given predictor?
These ratios can be requested by including estimate statements in the SAS code. For example, consider the following two estimate statements:

/* estimate ratios of expected counts associated with increases in given predictors */

proc genmod data = sasdata2.tecumseh;

   class sex marital1;

   model beer1 = sex age1 marital1 ed1 cig1 wtkg1 / 


dist = negbin link = log type3;

   estimate "Sex 1 vs. Sex 2 " sex 1 -1 / exp;

   estimate "One-year Education increase" ed1 1 / exp;

run;
Because SEX is declared as a categorical predictor in the class statement, we need to compare one level of SEX with another. To compare level 1 with level 2, we use the 

(1 -1) notation indicated in the code above, and exponentiate the coefficient representing the difference between the two levels. ED1 is a continuous predictor, so we simply need to specify a value of 1 after this predictor for SAS to exponentiate the estimated coefficient. Here are the results:

                                  Contrast Estimate Results

                                              Standard                                    Chi-

Label                              Estimate      Error    Alpha    Confidence Limits    Square

Sex 1 vs. Sex 2                      0.5636     0.0962     0.05     0.3750     0.7522    34.29

Exp(Sex 1 vs. Sex 2)                 1.7570     0.1691     0.05     1.4549     2.1217

One-year Education increase         -0.2884     0.0540     0.05    -0.3943    -0.1825    28.50

Exp(One-year Education increase)     0.7495     0.0405     0.05     0.6742     0.8332

                                     The GENMOD Procedure

                                   Contrast Estimate Results

                         Label                              Pr > ChiSq

                         Sex 1 vs. Sex 2                        <.0001

                         Exp(Sex 1 vs. Sex 2)

                         One-year Education increase            <.0001

                         Exp(One-year Education increase)
SAS includes 95% confidence intervals for the ratios of counts, based on the estimates and their standard errors. We see that the predicted mean count of BEER1 for SEX = 1 (males) is about 1.76 times that for females (controlling for the other variables in the model), while the predicted count decreases by about 25% with every one-year increase in education. Both of these variables are estimated to have a significant relationship with the BEER1 outcome.

The estimate statements in SAS are also very useful when considering categorical predictors with several categories, such as MARITAL1. For example, to estimate the ratio of expected BEER1 counts when comparing marital category 1 (married) to marital category 4 (divorced), we can use the following estimate statement (note that the two categories being compared are indicated with a 1 and a -1, and the other categories are assigned a 0):

proc genmod data = sasdata2.tecumseh;

   class sex marital1;

   model beer1 = sex age1 marital1 ed1 cig1 wtkg1 / 


dist = negbin link = log type3;

   estimate "Marital 1 vs. Marital 4" marital1 1 0 0 -1 / exp;

run;
                                 Contrast Estimate Results

                                            Standard                                    Chi-

  Label                          Estimate      Error    Alpha    Confidence Limits    Square

  Marital 1 vs. Marital 4         -0.1673     0.2635     0.05    -0.6837     0.3490     0.40

  Exp(Marital 1 vs. Marital 4)     0.8459     0.2229     0.05     0.5047     1.4177

                                     The GENMOD Procedure

                                   Contrast Estimate Results

                           Label                          Pr > ChiSq

                           Marital 1 vs. Marital 4            0.5253

                           Exp(Marital 1 vs. Marital 4)
Based on the estimated ratio (0.85) and its 95% confidence interval (0.51, 1.42), is there evidence of a significant difference between married and divorced respondents in terms of beer consumption, controlling for the other predictors in the model?

